Métodos

en

Biomatemática I:

Cálculo Integral

Y

Álgebra Lineal

con wxMaxima

Rafael Lahoz-Beltrá

METODOS EN BIOMATEMATICA I: CÁLCULO INTEGRAL Y ÁLGEBRA LINEAL

R. Lahoz-Beltra

Departamento de Matemática Aplicada (Biomatemática), Facultad de Biología,
C/ José Antonio Novais, 2
Universidad Complutense de Madrid. 28040 Madrid

Introducción

En este manual, y haciendo uso del sistema de álgebra computacional wxMaxima [1], se repasan aquellas técnicas de cálculo integral y conceptos de álgebra lineal en que se sustentan los modelos matemáticos en Biología. Se trata de una guía orientativa y eminentemente práctica en la que se asume que el lector está familiarizado con algunos conceptos elementales de matemáticas. Aunque hay abundante documentación en español sobre Maxima y wxMaxima, por ejemplo [2, 3], se recomienda el manual [4] por ser uno de los más completos, pudiendo descargarse gratuitamente desde la dirección indicada. Para algunos detalles y tareas más avanzadas sobre la programación en Maxima resulta de utilidad el documento web [5].

I. ¿Qué es Maxima? La interface wxMaxima

Maxima es un sistema de manipulación simbólica o Sistema de Algebra Computacional (CAS) con el que el usuario pude calcular derivadas, integrales, el polinomio de Taylor, ecuaciones diferenciales, sistemas de ecuaciones lineales, trabajar con polinomios, vectores, matrices, etc. Entre sus características destacaremos:

- Gran precisión de cálculo (incluye cálculos en coma flotante, *bfloat*).
- Representación de gráficos 2D y 3D basados en GNUPLOT.
- No desperdicia memoria principal del ordenador.
- Es multiplataforma, es decir hay versiones para distintos sistemas operativos, ya sea Windows, Linux o MacOS.
- Opciones de trabajo en modo *consola* o modo *windows* por medio de interfaces gráficas de usuario. Un ejemplo de interface gráfica multiplataforma es **wxMaxima**.
- El motor de cálculo simbólico está escrito en lenguaje Common Lisp (GNU/GPL).
- Maxima nace en 1998 y es un descendiente de una versión de Macsyma del año 1982, uno de los primeros CAS desarrollado en los años 60 y 70 del siglo XX en el MIT de EE.UU.
- Es un CAS de propósito general, y por tanto útil para cualquier clase de **cálculo simbólico** en problemas científico-tecnológicos y en cualquier área de conocimiento, por ejemplo en Biología, Economía, Física, etc.
- Se distribuye de forma gratuita bajo licencia ‘GNU General Public License’.



Pantalla de wxMaxima 0.7.1. (tomado de http://en.wikipedia.org/wiki/Maxima_(software))

La interface de usuario de wxMaxima es un **cuaderno** en el que haremos las operaciones, incluiremos anotaciones de texto, etc. organizándose en unidades de trabajo llamadas **celdas**:



Una vez escritas en las celdas las expresiones matemáticas según el lenguaje y sintaxis de Maxima, el contenido de las celdas es **evaluado**, es decir se obteniene un **OUTPUT %o** a partir del **INPUT %i** planteado. La evaluación del contenido se realiza a través de la ocpión **Celda** en la ventana de tareas de wxMaxima:



Por ejemplo, eligiendo la opción **Evaluar todas las celdas** obtendremos para cada celda **%i** el resultado correspondiente (**%o**):

(%i92) `(x^2+2*x-1)*(2*x^2-9*x+1);`

(%o92) $\left(x^2+2x-1\right)\left(2x^2-9x+1\right)$

(%i93) `expand((x^2+2*x-1)*(2*x^2-9*x+1));`

(%o93) $2x^4-5x^3-19x^2+11x-1$

(%i94) `factor(2*x^4-5*x^3-19*x^2+11*x-1);`

(%o94) $\left(x^2+2x-1\right)\left(2x^2-9x+1\right)$

(%i95) `x:5;`

(%o95) 5

(%i96) `(x^2+2*x-1)*(2*x^2-9*x+1);`

(%o96) 204

II. Métodos de cálculo integral útiles en Biomatemática

En esta sección estudiaremos las técnicas de integración que son de utilidad en Matemáticas, Estadística y Física. Utilizaremos wxMaxima para resolver las integrales, dejando como ejercicio su cálculo a mano. En general, una integral puede resolverse aplicando uno o varios de los siguientes métodos o **técnicas de integración** (**Caja 1**).

Caja 1

$$\begin{cases} \text{Integrales inmediatas} \begin{cases} \text{tipo logarítmico} \\ \text{“casi} - \text{inmediatas”} \end{cases} \\ \text{Integración por partes} \\ \text{Integrales de funciones racionales} \\ \text{Integración con cambio de variable} \end{cases}$$

II.1. Integrales inmediatas

Las integrales inmediatas más comunes se recogen en **tablas**, “estudiándose” de forma similar a la tabla de aminoácidos en Bioquímica, tal y como se muestra a continuación:

∫ K x dx = Kx + C

∫ u^n dx = u^n+1/n+1 + C

u^n . u' dx = u^n+1/n+1 + C

∫ 1/u dx = Ln(u) + C

u'/u dx = Ln(u) + C|

∫ a^u u' dx = a^u/Ln(a)+ C|

∫ e^u u' dx = e^u + C

∫ sin(u)dx = -cox(u) + C

sin(u)u' dx = -cos(u) + C

∫ cos(u)dx = sin(u) + C

cos(u)u‘ dx = sin(u) + C

∫ 1/1+u^2 u' dx = arctg(u) + C

∫ 1/sqrt(1-u^2) dx = arcsin(u) + C





En la tabla **u** es una función siendo **u’** su derivada. Con wxMaxima utilizaremos la orden **integrate** para resolver integrales. Ahora bien, si escribimos el apóstrofe (tecla con **?** en el teclado español) antes de la orden, es decir **‘integrate**, entonces únicamente se escribe la expresión con el símbolo integral, sin resolverse la expresión:

```
(%i1) 'integrate(3·x,x);
```

(%o1) $3\int x\,dx$

Una vez eliminemos el apóstrofe, la integral es calculada. Por ejemplo, resolver las siguientes integrales:

$$\int 6x^3\,dx \quad \int x^{2/3}\,dx \quad \int \frac{3}{x^4}\,dx \quad \int x^{1/3}\,dx$$

Se trata de algunos ejemplos de **integrales inmediatas** que resolveremos con wxMaxima:

```
(%i2) integrate(6*x^3, x);
```

(%o2) $\frac{3\,x^4}{2}$

```
(%i3) integrate(x^(2/3), x);
```

(%o3) $\frac{3\,x^{5/3}}{5}$

```
(%i4) integrate(3/(x^4),x);
```

(%o4) $-\frac{1}{x^3}$

```
(%i5) integrate(x^(1/3),x);
```

(%o5) $\frac{3\,x^{4/3}}{4}$

Otra clase de integrales inmediatas son de **tipo logarítmico**.

Por ejemplo, resolveremos las integrales que se muestran a continuación:

$$\int \frac{x^2}{x^3+6}\,dx \quad \int \frac{Cos\,x}{Sen\,x}\,dx \quad \int \frac{Sen(2x)}{1+Sen^2\,x}\,dx \quad \int \frac{1}{\sqrt{x}\,(1+\sqrt{x})}\,dx$$

Una vez más utilizaremos wxMaxima, recordando que wxMaxima no nos informa de la clase de integral de que se trata:

(%i6) integrate((x^2)/(x^3+6), x);

(%o6) $\dfrac{\log(x^3+6)}{3}$

(%i7) integrate(cos(x)/sin(x),x);

(%o7) $\log(\sin(x))$

(%i8) integrate(sin(2*x)/(1+sin(x)^2),x);

(%o8) $\log\left(\sin(x)^2+1\right)$

(%i9) integrate(1/(sqrt(x)*(1+sqrt(x))),x);

(%o9) $2\log(\sqrt{x}+1)$

En este grupo de las integrales inmediatas hay integrales a la que hemos denominado "casi-inmediatas". Considerando que los contenidos de este manual están orientados al tratamiento de modelos matemáticos en Biología [6], a partir de ahora utilizaremos la varible ***t***, el **tiempo**, en lugar de **x**.

Por ejemplo, la siguientes integrales ilustran esta clase de integrales inmediatas:

$$\int \frac{2t^3+t^2-t}{t^2}dt \quad \int \left(\frac{2}{5}\right)^t dt \quad \int t\, e^{t^2}\, dt \quad \int \frac{e^{-2t}+e^{2t}}{2}dt$$

Resolveremos con wxMaxima los ejemplos :

(%i10) integrate((2*t^3+t^2-t)/t^2,t);

(%o10) $\dfrac{2\,t^2+2\,t}{2}-\log(t)$

(%i11) ratsimp(integrate((2/5)^t,t));

(%o11) $-\dfrac{5^{\frac{\log(2)\,t}{\log(5)}-t}}{\log(5)-\log(2)}$

```
(%i12) integrate(t*%e^(t^2),t);
```

$$(\%o12)\ \frac{\%e^{t^2}}{2}$$

```
(%i13) integrate((exp(-2*t)+exp(2*t))/2,t);
```

$$(\%o13)\ \frac{\frac{\%e^{2\,t}}{2}-\frac{\%e^{-2\,t}}{2}}{2}$$

II.2. Integración por partes

La técnica de integración por partes, una de las más utilizadas en ciencias, consta de dos etapas. En primer lugar, se elige **u** y **dv**, siendo el paso más delicado ya que de dicha elección dependerá el éxito en la resolución de la integral. Por ello, aplicaremos la "regla ALPES" (**A**: funciones ciclométricas *arco*, por ej. arc sen(x), arc cos(x), arc tg(x), etc.; **L**: funciones logarítmicas, por ej. LOG(x), LN(t+1), etc.; **P**: polinomios, por ej. x, x^2+1, $x^3/3$, etc.; **E**: exponencial de una función, por ej. exp(x); **S**: en abreviatura de función sen(x) aunque pueda tratarse de cualquier otra función trigonométrica, por ej. cos(2x). Según esta regla nemotécnica "se etiqueta como **u** a aquella función que en la palabra ALPES esté ubicada a la izquierda con respecto de la otra función a la que designaremos **dv**". En segundo lugar, efectuaremos:

$$\int u.dv = u.v - \int v.du$$



Ilustración de la regla nemotécnica ALPES

Es importante recordar una vez más que wxMaxima no informa de la técnica de integración utilizada. Por ejemplo, si planteamos $\int (t-3)e^t\, dt$, una vez evaluada la expresión obtendremos:

```
(%i14) integrate ((t-3)*exp(t),t);
```

(%o14) $(t-1)\%e^{t}-3\ \%e^{t}$

Recordemos que si deseamos escribir la integral, añadiremos un apóstrofe al principio de la orden:

```
(%i15) 'integrate ((7*t^2-5)*sin(2*t),t);
```

(%o15) $\int (7\ t^2-5)\sin(2\ t)dt$

```
(%i16) 'integrate (exp(t)*cos(t),t);
```

(%o16) $\int \%e^{t}\cos(t)dt$

Acontinuación, mostraremos algunos ejemplos de integrales que se resuelven por esta técnica de integración:

$$\int t^3 e^t\,dt \quad \int (2t+4)e^{2t+4}\,dt \quad \int arctg(t)dt \quad \int \log(3t)dt \quad \int t^2\log(t)dt$$

resolviendose con wxMaxima:

```
(%i17) integrate (t^3*exp(t),t);
```

(%o17) $(t^3-3\ t^2+6\ t-6)\%e^{t}$

```
(%i18) integrate((2*t+4)*exp(2*t+4),t);
```

(%o18) $2\ \%e^{2\ t+4}+\frac{(2\ \%e^{4}\ t-\%e^{4})\%e^{2\ t}}{2}$

```
(%i19) integrate(atan(t),t);
```

$$\text{(\%o19)}\quad t\,\mathrm{atan}(t)-\frac{\log(t^2+1)}{2}$$

```
(%i20) integrate(log(3*t),t);
```

$$\text{(\%o20)}\quad \frac{3\,t\log(3\,t)-3\,t}{3}$$

```
(%i21) integrate((t^2)*log(t),t);
```

$$\text{(\%o21)}\quad \frac{t^3\log(t)}{3}-\frac{t^3}{9}$$

II.3. Integrales de funciones racionales

El cálculo de integrales de funciones racionales (cociente de dos polinomios) $\int \frac{p(x)}{q(x)}dx$ se resuelve "a mano" tal y como se explica a continuación. Explicaremos la resolución de diferentes casos:

Caso 3.1.- Sea $\int \frac{p(x)}{q(x)}dx$ una integral de una función racional en la que el cociente de los polinomios p(x)/q(x) cumple **p(x)** $\geq$ **q(x)**.

Por ejemplo, calcular $\int \frac{t^3}{t^2+1}dt$. Resolveremos la integral dividiendo entre sí los polinomios, y teniendo en cuenta que:

$$\frac{P}{Q}=C+\frac{R}{Q} \quad , \quad \frac{t^3}{t^2+1}=t+\frac{(-t)}{t^2+1}$$

Por consiguiente, la integral planteada será igual a:

$$\int \frac{t^3}{t^2+1}dt=\int t\,dt-\int \frac{t}{t^2+1}dt$$

siendo el resultado:

$$=\frac{1}{2}\left(t^2-\ln(t^2+1)+\mathrm{c}\right)$$

Caso 3.2.- Sea $\int \frac{p(x)}{q(x)} dx$ una integral de una función racional en la que el cociente de los polinomios p(x)/q(x) cumple **p(x)<q(x)**. La técnica de integración consiste en descomponer la integral original en otras que son integrales inmediatas, por lo general del tipo:

$$\int \frac{u'}{u} dx = \ln(u) + c \qquad \int \frac{u'}{1+u^2} dx = arctg(u) + c$$

siendo **u** una función y **u'** su derivada. Explicaremos la técnica de resolución con los siguientes ejemplos:

3.2.1.- Sea $\int \frac{t+1}{t^2-3t+2} dx$. En este caso la integral se descompone en dos integrales:

$$\int \frac{t+1}{t^2-3t+2} dt = \int \frac{t+1}{(t-1)(t-2)} dt = \int \frac{A}{t-1} dt + \int \frac{B}{t-2} dt$$

A continuación, realizamos con el fin de obtener los valores de A y B, las siguientes operaciones:

$$\begin{aligned} &t+1 = A(t-2) + B(t-1) \\ &(t-1) = 0, \quad t = 1, \quad 2 = -A \\ &(t-2) = 0, \quad t = 2, \quad 3 = B \end{aligned}$$

Finalmente, sustituyendo los valores de A y B se resuelven las integrales:

$$= -2\int \frac{1}{t-1} dt + 3\int \frac{1}{t-2} dt = -2\ln(t-1) + 3\ln(t-2) + C$$

Un ejemplo de este tipo de integral en Biomatemática es la que hay que resolver para obtener la solución general de la **ecuación diferencial de Verhust** o **ecuación logística** [6]:

$$y' = ry\left(1 - \frac{y}{K}\right)$$

siendo *r* la tasa de crecimiento y *K* la *capacidad de carga* en una población. Si sustituimos y' por dy/dt, operamos el término entre paréntesis, y reordenamos los términos tendremos que:

$$\frac{dy}{dt}=\frac{ry\left(K-y\right)}{K} \quad , \quad K.dy=ry(K-y).dt$$

Puesto que se trata de una ecuación diferencial de variables separables, entonces:

$$\frac{K}{r}\int_{y_0}^{y}\frac{1}{y(K-y)}dy=\int_{0}^{t}dt$$

La integral $\int_{y_0}^{y}\frac{1}{y(K-y)}dy$ se resuelve aplicando la técnica descrita en esta sección, es decir:

$$\int_{y_0}^{y}\frac{1}{y(K-y)}dy=\int_{y_0}^{y}\frac{A}{y}+\int_{y_0}^{y}\frac{B}{\left(K-y\right)}$$

siendo A=B=1/*K*.

3.2.2.- Sea $\int\frac{t^2}{(t+1)(t-1)^2}dt$. Puesto que hay un **término (t-1)^2** se realizará la siguiente descomposición de la integral, obteniendo los valores de A, B y C que sustituiremos, resolviéndose las integrales resultantes:

$$\int\frac{t^2}{(t+1)(t-1)^2}dt=\int\frac{A}{t+1}dt+\int\frac{B}{t-1}dt+\int\frac{C}{(t-1)^2}dt$$

$$t^2=A(t-1)(t-1)^2+B(t+1)(t-1)^2+C(t+1)(t-1)$$
$$t^2=A(t-1)^2+B(t+1)(t-1)+C(t+1)$$

$$(t-1)=0,\ t=1,\ ....\ C=1/2$$
$$(t+1)=0,\ t=-1,\ ...\ A=1/4$$
$$[t^2]\quad 1=A+B,...B=3/4$$

3.2.3.- Sea $\int\frac{t^2}{(t-2)(t^2+1)}dt$. En este ejemplo hay un **factor cuadrático (t^2-1)**, descomponiéndose la integral en otras dos integrales. A continuación, obtendremos los valores de A, B y C. Sustituyendo los valores obtenidos, resolveremos las integrales:

$$\int \frac{t^2}{(t-2)(t^2+1)}\,dt = \int \frac{A}{t-2}\,dt + \int \frac{Bt+C}{t^2+1}\,dt$$

$$t^2 = A(t^2+1) + (Bt+C)(t-2)$$

$$(t-2)=0, \quad t=2, \quad 4 = A5+0 \quad A = 4/5$$
$$[t^2] \quad 1 = A+B \quad B = 1/5$$
$$[0] \quad 0 = A-2C \quad C = A/2$$

Una forma de tratar integrales racionales con wxMaxima evitando **errores de escritura** consiste en escribir las expresiones de **p** y **q** de la función racional **p/q** por pasos.

Considérense los siguientes ejemplos:

$$\int \frac{3t^2+5t}{t^2-t-2}\,dt \qquad \int \frac{t+1}{t^2-3t+2}\,dt$$

```
(%i22) p:3*t^2+5*t;
(%o22) 3 t^2 +5 t

(%i23) q:t^2-t-2;
(%o23) t^2 -t -2

(%i24) integrate(p/q,t);
(%o24) 2 log(t+1)/3 + 22 log(t-2)/3 +3 t

(%i25) r:t+1;
(%o25) t+1

(%i26) s:t^2-3*t+2;
(%o26) t^2 -3 t +2

(%i27) integrate(r/s,t);
(%o27) 3 log(t-2)-2 log(t-1)
```

II.4. Integración con cambio de variable

La técnica del cambio de variable o **método de sustitución** efectúa un cambio de variable en el integrando por medio de una variable auxiliar **u**:

$$\int f(x)dx \text{ , sí } \mathbf{u}\text{=x entonces } \mathbf{du}\text{=}\mathbf{u'}\text{ dx}$$

se integra y posteriormente se deshace el cambio de variable. Ilustraremos la técnica de sustitución con el siguiente ejemplo.

Sea $\int (3x+2)^3\,dx$. Efectuando el cambio de variable:

$$\text{u=3x+2, du=3 dx, } \frac{\text{du}}{3}=\text{dx}$$

sustituyendo en la integral y operando, tendremos que:

$$\int u^3\,dx = \int \frac{u^3}{3}du = \frac{1}{3}\int u^3\,du = \frac{1}{3}.\frac{u^4}{4}+\text{c} = \frac{\left(3x+2\right)^4}{12}+\text{c}$$

A continuación, resolveremos con wxMaxima los siguientes ejemplos:

$$\int (2t+6)^5\,dt \quad \int \sqrt{4t-1}\,dt \quad \int e^{1-t}\,dt \quad \int t\,e^{t^2}\,dt$$

```
(%i28) integrate ((2*t+6)^5,t);
```

(%o28) $\frac{16\,t^6}{3}+96\,t^5+720\,t^4+2880\,t^3+6480\,t^2+7776\,t$

```
(%i29) integrate(sqrt(4*t-1),t);
```

(%o29) $\frac{(4\,t-1)^{3/2}}{6}$

```
(%i30) integrate(exp(1-t),t);
```

(%o30) $-\%e^{1-t}$

```
(%i31) integrate(t*exp(t^2),t);
```

(%o31) $\frac{\%e^{t^2}}{2}$

III. Elementos de algebra lineal en Biomatemática

Un vector *v* es un **objeto matemático** que se define como una colección de *n* números reales, tal que $v = (a_1, a_2, \ldots, a_n)$. El concepto de vector cuenta con numerosas aplicaciones en Biomatemática, tal y como se muestra en la figura.




steczki witaminy A:

to cząsteczka ta może być wyrażona za pomocą następującej macierzy:

$$\begin{pmatrix}
0&1&0&0&0&1&0&0&0&0&0&0&0&0\\
1&0&1&0&0&0&0&0&0&0&0&0&0&0\\
0&1&0&1&0&0&0&0&0&0&0&0&0&0\\
0&0&1&0&1&0&0&0&0&0&0&0&0&0\\
0&0&0&1&0&1&0&0&0&0&0&0&0&0\\
1&0&0&0&1&0&1&0&0&0&0&0&0&0\\
0&0&0&0&0&1&0&1&0&0&0&0&0&0\\
0&0&0&0&0&0&1&0&1&0&0&0&0&0\\
0&0&0&0&0&0&0&1&0&1&0&0&0&0\\
0&0&0&0&0&0&0&0&1&0&1&0&0&0\\
0&0&0&0&0&0&0&0&0&1&0&1&0&0\\
0&0&0&0&0&0&0&0&0&0&1&0&1&0\\
0&0&0&0&0&0&0&0&0&0&0&1&0&1\\
0&0&0&0&0&0&0&0&0&0&0&0&1&0
\end{pmatrix}$$

Przyjmujemy dla x_{ij} wartość 1, jeżeli atomy *i*, *j* są między sobą związane, i wartość 0 w przypadku przeciwnym, to znaczy gdy wiązanie między nimi nie istnieje.

87

Aplicaciones de los vectores y matrices en Biología. (Izquierda) Estudio de dinámica de poblaciones, ley de Hardy-Weinberg, matriz estocástica cuyos elementos son las probabilidades de transición o mutación entre bases del ADN y expresión para un sistema dinámico. (Derecha) Representación de la molécula de vitamina A (tomado de la edición en polaco del libro "R. Lahoz-Beltra. 2010. Las Matemáticas de la Vida. Modelos numéricos para la biología y la ecología. Colección El Mundo es Matemático. RBA")

A continuación, y asumiendo que el lector conoce qué es un espacio vectorial, los conceptos de dependencia e independencia lineal, qué es generador y una base, entre otros conceptos generales, abordaremos con **wxMaxima** el estudio elemental del **álgebra lineal**.

El álgebra lineal comprende el estudio de las principales operaciones con vectores, matrices y ecuaciones lineales. El álgebra lineal constituye junto con las **ecuaciones diferenciales** dos de los pilares esenciales en los que se sustenta la **Biología Matemática** o **Biomatemática**.

III.1. Dependencia e independencia lineal

¿Son los vectores (8, 2, 0), (1, 5, 1) y (0, 1, 1) linealmente dependientes o independientes?

```
(%i1) u:[8,2,0];
(%o1) [8,2,0]

(%i2) v:[1,5,1];
(%o2) [1,5,1]

(%i3) w:[0,1,1];
(%o3) [0,1,1]

(%i4) M:matrix(u,v,w);
```

(%o4) $\begin{bmatrix} 8 & 2 & 0 \\ 1 & 5 & 1 \\ 0 & 1 & 1 \end{bmatrix}$

```
(%i5) rank(M);
(%o5) 3
```

Son linealmente independientes ya que el rango es 3.

¿Y los vectores (1, -2, 1), (-2, 4, -2), (1, 0, 0)?

```
(%i6) u:[1,-2,1];
(%o6) [1,-2,1]

(%i7) v:[-2,4,-2];
(%o7) [-2,4,-2]

(%i8) w:[1,0,0];
(%o8) [1,0,0]

(%i9) M:matrix(u,v,w);
```

(%o9) $\begin{bmatrix} 1 & -2 & 1 \\ -2 & 4 & -2 \\ 1 & 0 & 0 \end{bmatrix}$

```
(%i10) rank(M);
(%o10) 2
```

Son linealmente dependientes ya que el rango es menor que 3. El vector v es múltiplo del vector u.

III.2. Sistema generador del espacio vectorial

Sean los vectores (1, 0, 0), (-2, 0, 1) y (1, 3, -2) de un subespacio vectorial S, tal que S V ¿constituyen los vectores un generador del espacio vectorial V?

(%i1) M:matrix([1,0,0],[-2,0,1],[1,3,-2]);

(%o1) $\begin{bmatrix} 1 & 0 & 0 \\ -2 & 0 & 1 \\ 1 & 3 & -2 \end{bmatrix}$

(%i2) C:[c1,c2,c3];

(%o2) $[c1, c2, c3]$

(%i3) C:transpose(C);

(%o3) $\begin{bmatrix} c1 \\ c2 \\ c3 \end{bmatrix}$

(%i4) V:[a,b,c];

(%o4) $[a, b, c]$

(%i5) V:transpose(V);

(%o5) $\begin{bmatrix} a \\ b \\ c \end{bmatrix}$

(%i6) M.C=V;

(%o6) $\begin{bmatrix} c1 \\ c3-2\ c1 \\ -2\ c3+3\ c2+c1 \end{bmatrix} = \begin{bmatrix} a \\ b \\ c \end{bmatrix}$

(%i7) RES:M.C-V;

(%o7) $\begin{bmatrix} c1-a \\ c3-2\ c1-b \\ -2\ c3+3\ c2+c1-c \end{bmatrix}$

(%i8) SISECS:list_matrix_entries(RES);

(%o8) $[c1-a, c3-2\ c1-b, -2\ c3+3\ c2+c1-c]$

(%i9) linsolve(SISECS,[a,b,c]);

(%o9) $[a=c1, b=c3-2\ c1, c=-2\ c3+3\ c2+c1]$

Efectivamente los vectores son un sistema generador del espacio vectorial V ya que el **sistema es compatible**, en el ejemplo **indeterminado**, siendo claramente al menos uno de los términos *a*, *b* o *c* distinto de cero.

III.3. Base del espacio vectorial

Los vectores (1, 0, 0), (0, 1, 0) y (0, 0, 1) ¿son una base del espacio vectorial *V*?

En primer lugar comprobaremos que son vectores linealmente independientes:

```
(%i1) M:matrix([1,0,0],[0,1,0],[0,0,1]);
```

$$(\%o1) \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
(%i2) rank(M);
(%o2) 3

(%i3) determinant(M);
(%o3) 1
```

Seguidamente, comprobaremos que son un sistema de generadores:

```
(%i4) C:[c1,c2,c3];
(%o4) [c1, c2, c3]

(%i5) C:transpose(C);
```

$$(\%o5) \begin{bmatrix} c1 \\ c2 \\ c3 \end{bmatrix}$$

```
(%i6) V:[a,b,c];
(%o6) [a, b, c]

(%i7) V:transpose(V);
```

$$(\%o7) \begin{bmatrix} a \\ b \\ c \end{bmatrix}$$

```
(%i8) M.C=V;
```

$$(\%o8)\quad \begin{bmatrix} c1 \\ c2 \\ c3 \end{bmatrix} = \begin{bmatrix} a \\ b \\ c \end{bmatrix}$$

```
(%i9) RES:M.C-V;
```

$$(\%o9)\quad \begin{bmatrix} c1-a \\ c2-b \\ c3-c \end{bmatrix}$$

```
(%i10) SISECS:list_matrix_entries(RES);
(%o10) [c1-a, c2-b, c3-c]

(%i11) linsolve(SISECS,[a,b,c]);
(%o11) [a=c1, b=c2, c=c3]
```

Los vectores si son una base del espacio vectorial V ya que son linealmente independientes, el rango es 3 y el determinante distinto de cero, y un sistema generador puesto que el **sistema es compatible** y **determinado**.

III.4. Cambio de base

Sea un vector v=(1, 1, 3), expresar el vector v en la base B1={(1, 1, 1), ((1, 1, 0), (1, 0, 0)} y en B2={(2, 1, 2), (1, 0, 3), (-1, 4, -2)}. Obtener las matrices de paso P y Q que permiten el cambio de base de B1 a B2 y de B2 a B1, respectivamente.

En primer lugar, definiremos la **base B1**:

```
(%i1) u1:[1,1,1];
(%o1) [1,1,1]

(%i2) u2:[1,1,0];
(%o2) [1,1,0]

(%i3) u3:[1,0,0];
(%o3) [1,0,0]

(%i4) B1:transpose(matrix(u1,u2,u3));
```

$$(\%o4)\quad \begin{bmatrix} 1 & 1 & 1 \\ 1 & 1 & 0 \\ 1 & 0 & 0 \end{bmatrix}$$

y en segundo lugar, la **base B2**:

```
(%i5) w1:[2,1,2];
(%o5) [2,1,2]

(%i6) w2:[1,0,3];
(%o6) [1,0,3]

(%i7) w3:[-1,4,-2];
(%o7) [-1,4,-2]

(%i8) B2:transpose(matrix(w1,w2,w3));
```

$$(\%o8) \begin{bmatrix} 2 & 1 & -1 \\ 1 & 0 & 4 \\ 2 & 3 & -2 \end{bmatrix}$$

En tercer lugar, el **vector v**:

```
(%i9) V:[1,1,3];
(%o9) [1,1,3]

(%i10) V:transpose(V);
```

$$(\%o10) \begin{bmatrix} 1 \\ 1 \\ 3 \end{bmatrix}$$

¿Cómo obtendremos las coordenadas del vector **v** en la base **B1**:

```
(%i11) C:[c1,c2,c3];
(%o11) [c1,c2,c3]

(%i12) C:transpose(C);
```

$$(\%o12) \begin{bmatrix} c1 \\ c2 \\ c3 \end{bmatrix}$$

```
(%i13) B1.C;
```

(%o13) $\begin{bmatrix} c3+c2+c1 \\ c2+c1 \\ c1 \end{bmatrix}$

```
(%i14) B1.C=V;
```

(%o14) $\begin{bmatrix} c3+c2+c1 \\ c2+c1 \\ c1 \end{bmatrix} = \begin{bmatrix} 1 \\ 1 \\ 3 \end{bmatrix}$

```
(%i15) SISECS:list_matrix_entries(B1.C-V);
```

(%o15) $[c3+c2+c1-1, c2+c1-1, c1-3]$

```
(%i16) linsolve(SISECS,[c1,c2,c3]);
```

(%o16) $[c1=3, c2=-2, c3=0]$

¿Y las coordenadas de **v** en **B2**:

```
(%i17) B2;
```

(%o17) $\begin{bmatrix} 2 & 1 & -1 \\ 1 & 0 & 4 \\ 2 & 3 & -2 \end{bmatrix}$

```
(%i18) C;
```

(%o18) $\begin{bmatrix} c1 \\ c2 \\ c3 \end{bmatrix}$

```
(%i19) B2.C;
```

(%o19) $\begin{bmatrix} -c3+c2+2\ c1 \\ 4\ c3+c1 \\ -2\ c3+3\ c2+2\ c1 \end{bmatrix}$

```
(%i20) B2.C=V;
```

$$\text{(\%o20)} \quad \begin{bmatrix} -c3+c2+2\ c1 \\ 4\ c3+c1 \\ -2\ c3+3\ c2+2\ c1 \end{bmatrix} = \begin{bmatrix} 1 \\ 1 \\ 3 \end{bmatrix}$$

```
(%i21) SISECS2:list_matrix_entries(B2.C-V);
```

$$\text{(\%o21)} \quad [-c3+c2+2\ c1-1, 4\ c3+c1-1, -2\ c3+3\ c2+2\ c1-3]$$

```
(%i22) linsolve(SISECS2,[c1,c2,c3]);
```

$$\text{(\%o22)} \quad [c1=\frac{1}{17}, c2=\frac{19}{17}, c3=\frac{4}{17}]$$

Finalmente, concluiremos la sesión de trabajo obteniendo la **matriz de paso P** de **B1** a **B2**:

```
(%i61) u1:transpose(u1);
```

$$\text{(\%o61)} \quad \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix}$$

```
(%i62) B2.C=u1;
```

$$\text{(\%o62)} \quad \begin{bmatrix} -c3+c2+2\ c1 \\ 4\ c3+c1 \\ -2\ c3+3\ c2+2\ c1 \end{bmatrix} = \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix}$$

```
(%i63) SISECS2:list_matrix_entries(B2.C-u1);
```

$$\text{(\%o63)} \quad [-c3+c2+2\ c1-1, 4\ c3+c1-1, -2\ c3+3\ c2+2\ c1-1]$$

```
(%i64) linsolve(SISECS2,[c1,c2,c3]);
```

$$\text{(\%o64)} \quad [c1=\frac{9}{17}, c2=\frac{1}{17}, c3=\frac{2}{17}]$$

(%i65) `u2:transpose(u2);`

(%o65) $\begin{bmatrix} 1 \\ 1 \\ 0 \end{bmatrix}$

(%i66) `B2.C=u2;`

(%o66) $\begin{bmatrix} -c3+c2+2\ c1 \\ 4\ c3+c1 \\ -2\ c3+3\ c2+2\ c1 \end{bmatrix} = \begin{bmatrix} 1 \\ 1 \\ 0 \end{bmatrix}$

(%i67) `SISECS2:list_matrix_entries(B2.C-u2);`

(%o67) $[-c3+c2+2\ c1-1, 4\ c3+c1-1, -2\ c3+3\ c2+2\ c1]$

(%i68) `linsolve(SISECS2,[c1,c2,c3]);`

(%o68) $[c1=\frac{13}{17}, c2=-\frac{8}{17}, c3=\frac{1}{17}]$

(%i69) `u3:transpose(u3);`

(%o69) $\begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix}$

(%i70) `B2.C=u3;`

(%o70) $\begin{bmatrix} -c3+c2+2\ c1 \\ 4\ c3+c1 \\ -2\ c3+3\ c2+2\ c1 \end{bmatrix} = \begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix}$

(%i71) `SISECS2:list_matrix_entries(B2.C-u3);`

(%o71) $[-c3+c2+2\ c1-1, 4\ c3+c1, -2\ c3+3\ c2+2\ c1]$

(%i72) `linsolve(SISECS2,[c1,c2,c3]);`

(%o72) $[c1=\frac{12}{17}, c2=-\frac{10}{17}, c3=-\frac{3}{17}]$

siendo la matriz de paso **P**:

```
(%i73) P:matrix([9,13,12],[1,-8,-10],[2,1,-3]);
```

$$(\%o73)\quad \begin{bmatrix} 9 & 13 & 12 \\ 1 & -8 & -10 \\ 2 & 1 & -3 \end{bmatrix}$$

```
NOTA: producto 1/17·P con operador "asterisco":
```

```
(%i74) P:1/17·P;
```

$$(\%o74)\quad \begin{bmatrix} \frac{9}{17} & \frac{13}{17} & \frac{12}{17} \\ \frac{1}{17} & -\frac{8}{17} & -\frac{10}{17} \\ \frac{2}{17} & \frac{1}{17} & -\frac{3}{17} \end{bmatrix}$$

Obtendremos para concluir la sesión de trabajo la matriz de paso **Q** de **B2** a **B1**:

```
(%i75) Q:invert(P);
```

$$(\%o75)\quad \begin{bmatrix} 2 & 3 & -2 \\ -1 & -3 & 6 \\ 1 & 1 & -5 \end{bmatrix}$$

III.5. Álgebra de matrices

Sean las matrices A y B:

$$A = \begin{pmatrix} 2 & 5 & 8 \\ 1 & 2 & 1 \\ 6 & 9 & 4 \end{pmatrix} \qquad B = \begin{pmatrix} 1 & 3 & 7 \\ 4 & 9 & 0 \\ 2 & 1 & 3 \end{pmatrix}$$

Estudiaremos con wxMaxima las siguientes órdenes y **operaciones algebraicas**:

a) ¿Qué tarea realizan las órdenes **ident(3**) y **zeromatrix(2,3)**.
b) Efectúe las siguientes operaciones: A+(A-1), A*B y A^3.
c) Obtenga: transpuesta de A, inversa de A, adjuntos de A.
d) ¿Qué tarea realiza la orden **minor(A,1,2)**?
e) Obtenga el determinante de A y su rango.

Escribiremos en wxMaxima las dos matrices A y B, tal que $M_n x_m$ (n=filas, m=columnas):

```
(%i1) A:matrix([2,5,8],[1,2,1],[6,9,4]);
```

$$(\%o1) \begin{bmatrix} 2 & 5 & 8 \\ 1 & 2 & 1 \\ 6 & 9 & 4 \end{bmatrix}$$

```
(%i2) B:matrix([1,3,7],[4,9,0],[2,1,3]);
```

$$(\%o2) \begin{bmatrix} 1 & 3 & 7 \\ 4 & 9 & 0 \\ 2 & 1 & 3 \end{bmatrix}$$

a) Matrices especiales: identidad(tamaño) y nula(n, m):

```
(%i3) ident(3);
```

$$(\%o3) \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
(%i4) zeromatrix(2,3);
```

$$(\%o4) \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

b) Operaciones con matrices:

En primer lugar calcularemos A+(A-1). En segundo lugar, realizaremos el producto de A por B, teniendo en cuenta que en wxMaxima hay dos operadores: "punto abajo" A.B o producto estándar y "punto arriba" A*B (se escribe con asterisco) multiplicándose elemento a elemento. En tercer, último lugar, realizaremos la operación potencia utilizando para ello el operador **^**, por ejemplo A^3:

```
(%i5) A+(A-1);
```

$$(\%o5) \begin{bmatrix} 3 & 9 & 15 \\ 1 & 3 & 1 \\ 11 & 17 & 7 \end{bmatrix}$$

```
(%i6) A*B;
```

$$(\%o6) \begin{bmatrix} 2 & 15 & 56 \\ 4 & 18 & 0 \\ 12 & 9 & 12 \end{bmatrix}$$

```
(%i7) A^3;
```

$$(\%o7) \begin{bmatrix} 8 & 125 & 512 \\ 1 & 8 & 1 \\ 216 & 729 & 64 \end{bmatrix}$$

c) Operaciones especiales del álgebra de matrices:

Mostraremos a continuación cómo obtener la transpuesta, inversa o inversa dejando el valor del determinante fuera de una matriz, y el adjunto de una matriz dada:

```
(%i8) transpose(A);
```

$$(\%o8) \begin{bmatrix} 2 & 1 & 6 \\ 5 & 2 & 9 \\ 8 & 1 & 4 \end{bmatrix}$$

```
(%i9) invert(A);
```

$$(\%o9) \begin{bmatrix} \frac{1}{16} & -\frac{13}{4} & \frac{11}{16} \\ -\frac{1}{8} & \frac{5}{2} & -\frac{3}{8} \\ \frac{3}{16} & -\frac{3}{4} & \frac{1}{16} \end{bmatrix}$$

```
(%i10) invert(A),detout;
```

$$(\%o10) -\frac{\begin{bmatrix} -1 & 52 & -11 \\ 2 & -40 & 6 \\ -3 & 12 & -1 \end{bmatrix}}{16}$$

d) Menor:

Obtendremos el menor de la matriz A que resulta de cortar la primera fila y segunda columna.

```
(%i13) minor(A,1,2);
```

$$(\%o13)\quad \begin{bmatrix} 1 & 1 \\ 6 & 4 \end{bmatrix}$$

e) Calcularemos el determinante y rango de la matriz A:

```
(%i13) determinant(A);
(%o13) −16

(%i14) rank(A);
(%o14) 3
```

En una matriz cuadrada M_{nxn} (número de filas=número de columnas) si se cumple que el determinante de M es igual a cero entonces los vectores de M son **linealmente dependientes** si el rango de la matriz M es menor que *n*, es decir en wxMaxima sí **determinant(M)**=0 y **rank(M)**<n. Si el determinante de M es distinto de cero entonces los vectores son **linealmente independientes** si el rango de M es igual a *n* (**determinant(M)** $\neq$ 0 y **rank(M)**=n).

III.6. Modelos matriciales iterativos. Ejemplo de aplicación lineal

El concepto de **transformación lineal T**: $\mathbb{R}^n \to \mathbb{R}^n$ en el álgebra lineal (véase [7]) es "equivalente" al concepto de proporcionalidad en una ecuación diferencial y'=*k*.y o en una función f(x)=*k*.x. Por consiguiente, un **vector original u** puede ser transformado en un **vector imagen T(u)** si es multiplicado por una **matriz *A*** llamada **matriz de estados**, tal que T(u)=*A*.u. En la práctica las transformaciones lineales cuentan con numerosas aplicaciones prácticas al tratarse de **transformaciones geométricas**, siendo el vector T(u) una **proyección** (1), **rotación** (2) u **homotecia** (multiplicamos una distancia por un número *k*, caso (3)) dependiendo de que la matriz *A* sea por ejemplo y en cada caso (1, 2):

$$(1)\quad T(u) = \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix} .u \qquad (2)\quad T(u) = \begin{pmatrix} Cos\,\alpha & -Sen\,\alpha \\ Sen\,\alpha & Cos\,\alpha \end{pmatrix} .u \qquad (3)\quad T(u) = k.u$$

En Biomatemática la utilidad de las transformaciones lineales es tal que son la herramienta que permite al biólogo estudiar la evolución o **comportamiento dinámico** de poblaciones en **ecología**:

$$\begin{pmatrix} N_1(t+1) \\ N_2(t+1) \end{pmatrix} = A \begin{pmatrix} N_1(t) \\ N_2(t) \end{pmatrix}$$

En el caso particular de que los vectores T(u) sean **vectores nulos** {0}:

$$\begin{pmatrix} 0 \\ 0 \end{pmatrix} = A \begin{pmatrix} N_1(t) \\ N_2(t) \end{pmatrix}$$

entonces se conoce como **núcleo** o **kernel** o **Ker** de la transformación lineal a todos aquellos vectores **u** que tienen como vectores imagen **T(u)** a los vectores nulos. Estudiaremos una aplicación en biología de los conceptos que hemos expuesto.

Supóngase el siguiente experimento. En una charca un equipo de biólogos toma en un recipiente una muestra de agua haciendo un recuento con el microscopio en el laboratorio, siendo las condiciones iniciales N1(0)=5 y N2(0)=8 ejemplares de dos especies de protozoos, *Stylonichia* y *Ameba*, respectivamente.



También se obtuvo en el laboratorio la matriz de paso para las poblaciones de estas dos especies:

$$A = \begin{pmatrix} 1 & 1 \\ 0.75 & 0 \end{pmatrix}$$

Estudiaremos con wxMaxima la evolución a lo largo del tiempo de los efectivos N1 y N2 de las dos poblaciones de protozoos, **iterando** la transformación lineal N(t+1)=A.N(t) a lo largo de 20 unidades de tiempo:

```
(%i1) N1[0]:5;
      N2[0]:8;
(%o1) 5
(%o2) 8

(%i3) N[0]:transpose([N1[0],N2[0]]);
```

$$(\%o3) \quad \begin{bmatrix} 5 \\ 8 \end{bmatrix}$$

```
(%i4) A:matrix([1,1],[0.75,0]);
```

$$(\%o4) \quad \begin{bmatrix} 1 & 1 \\ 0.75 & 0 \end{bmatrix}$$

Observe cómo aplicando el **método iterativo *mapeamos*** el valor de **N** desde el tiempo ***t*** a ***t*+1**, con la siguiente rutina en wxMaxima, **prediciendo** los tamaños futuros de las dos poblaciones:

```
(%i5) for t:0 thru 20 do block(
      N[t+1]:A.N[t],
      N[t]:N[t+1],
      EQ:row(N[t],1)/row(N[t],2),
      print(N[t],EQ)

      );
```

Llamaremos **EQ** al cociente entre el número de individuos N1 de la primera especie y el número de individuos N2 de la segunda especie. La finalidad por la que calculamos este cociente es con la idea de estudiar si hay o no una tendencia a mantenerse constante o en **equilibrio** la proporción de efectivos ente ambas poblaciones.

Realizado el experimento, obtenemos para cada valor de tiempo ***t*** el vector $\begin{pmatrix} N_1(t) \\ N_2(t) \end{pmatrix}$ y el valor EQ:

$$\begin{bmatrix} 13 \\ 3.75 \end{bmatrix} \begin{bmatrix} 3.466666666666667 \end{bmatrix}$$

$$\begin{bmatrix} 16.75 \\ 9.75 \end{bmatrix} \begin{bmatrix} 1.717948717948718 \end{bmatrix}$$

…

$$\begin{bmatrix} 26.5 \\ 12.5625 \end{bmatrix} \begin{bmatrix} 2.109452736318408 \end{bmatrix}$$

$$\begin{bmatrix} 39.0625 \\ 19.875 \end{bmatrix} \begin{bmatrix} 1.965408805031447 \end{bmatrix}$$

$$\begin{bmatrix} 58.9375 \\ 29.296875 \end{bmatrix} \begin{bmatrix} 2.011733333333333 \end{bmatrix}$$

$$\begin{bmatrix} 88.234375 \\ 44.203125 \end{bmatrix} \begin{bmatrix} 1.996111700247437 \end{bmatrix}$$

$$\begin{bmatrix} 132.4375 \\ 66.17578125 \end{bmatrix} \begin{bmatrix} 2.001298624638451 \end{bmatrix}$$

$$\begin{bmatrix} 198.61328125 \\ 99.328125 \end{bmatrix} \begin{bmatrix} 1.999567406009124 \end{bmatrix}$$

$$\begin{bmatrix} 297.94140625 \\ 148.9599609375 \end{bmatrix} \begin{bmatrix} 2.0001442291933 \end{bmatrix}$$

$$\begin{bmatrix} 446.9013671875 \\ 223.4560546875 \end{bmatrix} \begin{bmatrix} 1.999951927068993 \end{bmatrix}$$

$$\begin{bmatrix} 670.357421875 \\ 335.176025390625 \end{bmatrix} \begin{bmatrix} 2.000016024695512 \end{bmatrix}$$

$$\begin{bmatrix} 1005.533447265625 \\ 502.76806640625 \end{bmatrix} \begin{bmatrix} 1.999994658477627 \end{bmatrix}$$

Un procedimiento alternativo de estudiar el comportamiento dinámico de las dos poblaciones consiste en **diagonalizar** la matriz de estados ***A***:

```
Método consistente en diagonalizar la matriz A

(%i6) eigenvalues(A);
rat: replaced -0.75 by -3/4 = -0.75
(%o6) [[3/2,-1/2],[1,1]]
```

Observe que lo que se conoce como **valor propio**, 3/2 en el experimento, y por tanto 1.5, indica que podríamos escribir:

$$\begin{pmatrix} N_1(t+1) \\ N_2(t+1) \end{pmatrix} = 1.5 \begin{pmatrix} N_1(t) \\ N_2(t) \end{pmatrix}$$

Concluyéndose que a lo largo del tiempo el número de individuos N1 de *Stylonichia* se "impondrá" numéricamente al número o efectivo N2 de amebas.

III.7. Diagonalización de matrices

Diagonalizar una matriz **A** consiste en encontrar vectores invariantes **v**, llamados **vectores propios** o **eigenvectores** tal que al multiplicar la matriz A por dicho vector v sólo cambie su magnitud –esto es, estirándose o encogiéndose λ unidades de longitud- y no su dirección, aunque pueda hacerlo su sentido. Por consiguiente tendremos que:

$$\mathbf{A} \,.\, \mathbf{v} = \lambda \,.\, \mathbf{v}$$

conociéndose al escalar λ con el nombre de **valor propio** o **eigenvalor**. La diagonalización de una matriz A "a mano" pasa por obtener las soluciones o raíces λ_i del llamado **polinomio característico** a partir del determinante de |A - λ.I| = 0, siendo I la matriz identidad. En wxMaxima el polinomio característico, los valores propios y los vectores propios se obtienen con las órdenes **charpoly**, **eigenvalues**, **eigenvectors**, respectivamente.

Supóngase que **diagonalizamos** con wxMaxima la siguiente matriz A, obteniendo sus **valores propios** y **vectores propios**:

```
(%i1) A:matrix([2,2,1],[-1,-1,-1],[2,4,3]);
```

$$(\%o1)\quad \begin{bmatrix} 2 & 2 & 1 \\ -1 & -1 & -1 \\ 2 & 4 & 3 \end{bmatrix}$$

```
(%i2) charpoly(A,x);
```

$$(\%o2)\quad -2(x-1)+((-x-1)(3-x)+4)(2-x)-2(-x-1)-4$$

```
(%i3) factor(%);
```

$$(\%o3)\quad -(x-2)(x-1)^2$$

```
(%i4) eigenvalues(A);
(%o4) [[2,1],[1,2]]

(%i5) eigenvectors(A);
(%o5) [[[2,1],[1,2]],[[[1,-1,2]],[[1,0,-1],[0,1,-2]]]]
```

Comprobaremos a continuación si la matriz A es diagonalizable:

```
(%i6) P:transpose(matrix([1,-1,2],[1,0,-1],[0,1,-2]));
```

$$(\%o6)\quad \begin{bmatrix} 1 & 1 & 0 \\ -1 & 0 & 1 \\ 2 & -1 & -2 \end{bmatrix}$$

```
(%i7) D:matrix([2,0,0],[0,1,0],[0,0,1]);
```

$$(\%o7)\quad \begin{bmatrix} 2 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
(%i8) rank(P);
(%o8) 3

(%i9) determinant(P);
(%o9) 1

(%i10) invert(P).A.P;
```

$$(\%o10)\quad \begin{bmatrix} 2 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
(%i11) P.D.invert(P);
```

$$(\%o11)\quad \begin{bmatrix} 2 & 2 & 1 \\ -1 & -1 & -1 \\ 2 & 4 & 3 \end{bmatrix}$$

Sea la matriz A:

$$\begin{pmatrix} 2 & 2 \\ 2 & -1 \end{pmatrix}$$

Obtenidos los valores y vectores propios ¿cuál es el **significado geométrico** de la diagonalización?



Si una matriz es diagonalizable entonces se cumple que **P^{-1}. A . P = D**, lo que puede ser comprobado utilizando la siguiente sentencia condicional en wxMaxima:

```
(%i12) if (invert(P).A.P)=D then print("DIAGONALIZABLE") else print ("NO DIAGONALIZABLE");
```

obteniendo la respuesta:

DIAGONALIZABLE

(%o12) *DIAGONALIZABLE*

Otra vía para comprobar si la matriz A con la que estamos trabajando es diagonalizable o lo que es equivalente ¿son los vectores asociados a un autovalor linealmente independientes? consiste en comprobar si la **multiplicidad algebraica** es igual a la **multiplicidad geométrica**, esto es si se cumple que **n-rango(A- λ I)**, tal y como se efectúa con wxMaxima en %i17 en el ejemplo:

```
(%i13) n:3; lambda:1; multialg:2;
(%o13) 3
(%o14) 1
(%o15) 2

(%i16) multigeom:n-rank(A-lambda·ident(n));
(%o16) 2


(%i17) if multialg=multigeom then print("DIAGONALIZABLE") else print ("NO DIAGONALIZABLE");
```

obteniendo, por segunda vez, la misma respuesta que en %o12:

```
DIAGONALIZABLE
 (%o17) DIAGONALIZABLE
```

Repetiremos el procedimiento anterior pero con una matriz A que **no es diagonalizable**:

```
(%i18) A:matrix([3,0,-1],[0,3,0],[1,-2,1]);
```

(%o18) $\begin{bmatrix} 3 & 0 & -1 \\ 0 & 3 & 0 \\ 1 & -2 & 1 \end{bmatrix}$

```
(%i19) charpoly(A,x);
```

(%o19) $-x+(1-x)(3-x)^2+3$

```
(%i20) factor(%);
```

(%o20) $-(x-3)(x-2)^2$

```
(%i21) eigenvalues(A);
(%o21) [[3,2],[1,2]]

(%i22) eigenvectors(A);
```

(%o22) $[[[3,2],[1,2]],[[[1,\frac{1}{2},0]],[[1,0,1]]]]$

Preguntándonos a continuación si la matriz en cuestión es o no diagonalizable:

```
(%i23) n:3; lambda:2; multialg:2;
(%o23) 3
(%o24) 2
(%o25) 2

(%i26) multigeom:n-rank(A-lambda·ident(n));
(%o26) 1
```

evaluando wxMaxima:

```
(%i27) if multialg=multigeom then print("DIAGONALIZABLE") else print ("NO DIAGONALIZABLE");
NO DIAGONALIZABLE
(%o27) NO DIAGONALIZABLE
```

Finalmente, diagonalizaremos una matriz A en la que los **valores propios** son **números complejos**:

(%i1) A:matrix([0,-2,-2],[4,-2,2],[4,-2,2]);

(%o1) $\begin{bmatrix} 0 & -2 & -2 \\ 4 & -2 & 2 \\ 4 & -2 & 2 \end{bmatrix}$

(%i2) charpoly(A,x);

(%o2) $-((-x-2)(2-x)+4)x+2(4(2-x)-8)-2(-4(-x-2)-8)$

(%i3) factor(%);

(%o3) $-x(x^2+16)$

(%i4) eigenvalues(A);

(%o4) [[-4 %i, 4 %i, 0], [1, 1, 1]]

(%i5) eigenvectors(A);

(%o5) [[[-4 %i, 4 %i, 0], [1, 1, 1]], [[[1, %i, %i]], [[1, -%i, -%i]], [[1, 1, -1]]]]

Las operaciones con **números complejos** [8] con wxMaxima requieren de las órdenes y sentencias que se muestran en la **Caja 2**.

Caja 2

Realizar en wxMaxima las siguientes tareas: (a) escribir $i = \sqrt{-1}$ que wxMaxima representa como %i, (b) escribir $\sqrt{-4}$, (c) resolver la ecuación $x^2+4=0$, (d) escribir el número complejo 2 + 5 i, (e) extraer la parte real de (d), (f) extraer la parte imaginaria de (d), obtener el valor absoluto o *módulo* del número complejo (d):

```
(%i6) sqrt(-1);
(%o6) %i

(%i7) sqrt(-4);
(%o7) 2 %i

(%i8) solve(x^2+4=0,x);
(%o8) [x=-2 %i, x=2 %i]

(%i9) numcomplejo: 2+ 5·%i;
(%o9) 5 %i+2

(%i10) realpart(numcomplejo);
(%o10) 2

(%i11) imagpart(numcomplejo);
(%o11) 5

(%i12) abs(numcomplejo);
(%o12) √29
```

A continuación se muestra cómo realizar operaciones aritméticas (+, -, * y /) con números complejos, simplificando el resultado de la división con la orden **ratsimp**:

```
(%i13) (2+5·%i)+(1-%i);
(%o13) 4 %i+3

(%i14) (2+5·%i)-(1-%i);
(%o14) 6 %i+1

(%i15) (2+5·%i)·(1-%i);
(%o15) (1-%i)(5 %i+2)

(%i16) -(2+2·%i)/(-2+2·%i);
```

$$(\%o16)\quad \frac{-2\,\%i-2}{2\,\%i-2}$$

```
(%i17) ratsimp(%);
```

$$(\%o17)\quad -\frac{\%i+1}{\%i-1}$$

Bibliografía

[1] A. Vodopivec et al. 2011. wxMaxima. http://wxmaxima.sourceforge.net/

[2] VVAA. 2010. Prácticas de Cálculo con wxMaxima. Universidad de Oviedo. Escuela Politécnica de Ingeniería de Gijón. http://www.unioviedo.es/bayon/calculo/Manual_wxMaxima.pdf

[3] VVAA. Prácticas de Ordenador con wxMaxima. Universidad de Granada. http://euler.us.es/~renato/clases/maxima/manualesPDF/maxima-manual-UGR.pdf

[4] R. Ipanaqué Chero. 2012. Breve Manual de Maxima (2ª Edición). http://www.unp.edu.pe/pers/ripanaque/download/manual.pdf

[5] J.M. Mira Ros. 2011. Elementos para prácticas con Maxima. http://webs.um.es/mira/maxima/manualico.php

[6] R. Lahoz-Beltra. 2014. Métodos en Biomatemática II: Ecuaciones Diferenciales con wxMaxima. eprints.ucm.es/26851/

[7] M. A. Martín. 2013. Matemáticas Bioenriquecidas. Ed. del autor. Véase referencia al libro en la página web: http://www.matematicasbioenriquecidas.com/

[8] R. Lahoz-Beltra. 2010. Las Matemáticas de la Vida. Modelos numéricos para la biología y la ecología. Colección "El Mundo es Matemático" RBA.